         **  STRINGS AND ERRORS **PRIVATE 

     FORMAT:  TIME (T)

     EXAMPLE: 95 ? TIME/60

     EXAMPLE #2:

     10 CLS:?:?

     20 ? "TIMING A LOOP":?:?

     30 S=TIME:REM STARTING TIME

     40 FOR X=1 TO 120:? X;", ";:

        NEXT X

     50 E=(TIME-S)/60:REM E=ELAPSED,

        TIME-S = STARTING TIME

     60 ?:?:? INT(E*100)/100;

        " SECONDS"

     The TIME counter is active as soon as you turn on your computer. It counts 60 numbers per second. To get a readout in seconds use TIME/60. To get a readout in minutes use TIME/3600. To get a readout in hours use TIME/216000. The TIME counter will continue as long as your computer is turned on. If you want to start the time counter from the beginning then press the RESET

button.

     FORMAT:  TIME$ (T)

     EXAMPLE: 30 ? TIME$

     EXAMPLE #2:

     10 CLS:?:?

     20 DIM T$(6)

     30 INPUT "SET TIME - HHMMSS";

        T$

     40 INPUT "WHEN TO STOP - HHMMSS";S

     50 CLS

     60 TIME$=T$

     70 REPEAT

     80 POKE 755,0:REM NO CURSOR

     90 POSITION 10,10:? "HOURS,

        MIN,SEC:";:? TIME$

     100 X=VAL(TIME$)

     110 UNTIL X=S

     120 POSITION 10,12:? "RING!

         RING! RING!"

     130 POKE 755,3:REM CURSOR

This command prints the time of day in the form of HOUR MINUTE SECOND (HHMMSS). In Europe they use 50 cycle current and European ATARI'S run on this current as against 60 cycle in the USA.  As a result of this, the TIME$ in unmodified versions of Turbo Basic function is not accurate and gains 10 minutes each hour.  In a modified version, TIME$ keeps accurate time.

     FORMAT:  INKEY$ (T)

     EXAMPLE: 70 ? INKEY$

              90 A$=INKEY$

     EXAMPLE #2

     10 REM SCREEN WRITER

     20 ? INKEY$;

     30 GOTO 20

     EXAMPLE #3

     10 CLS:DIM A$(1):?:?:REM GAME

     20 ? "HIT THE KEY OF THE LETTER ON      THE SCREEN!"

     30 PAUSE 150

     40 R=RAND(90):IF R<65 THEN 40

     50 CLS:POSITION 19,12:

        ? CHR$(R)

     60 PAUSE 55

     70 A$=INKEY$

     80 IF A$=CHR$(R):GOTO 90:

        ELSE:GOTO 100:ENDIF

     90 ?:? "YOU GOT IT!":PAUSE 90:

        GOTO 40

     100 ?:? "YOU MISSED IT!":

        PAUSE 90:GOTO 40

     This command returns the character of the last key hat was pressed. In example #3 please note line 70. Line 80 uses and IF-ELSE-ENDIF command.

     FORMAT:  INSTR(x$,a$,i) (T)

     EXAMPLE: 50 ? INSTR(X$,A$,1)

     EXAMPLE #2:

     10 DIM X$(30),A$(5)

     20 X$="MY HEART BELONGS TO

        DADDY"

     30 A$="DADDY"

     40 ? "DADDY IS AT POSIT. ";

     50 ? INSTR(X$,A$,1)

     This command returns the beginning of a sub string. The x$ = the original string. The a$ = the sub string you want to find. The i = the position in numeric form that you want to search the original string.  This command will return the position number where the sub string starts.  Spaces are also counted.

     FORMAT:  UINSTR(x$,a$) (T)

              UINSTR(x$,a$,i)

     EXAMPLE: 50 ? UINSTR(X$,A$)

              20 ? UNISTR(X$,A$,5)

     EXAMPLE #2

     10 DIM X$(40),A$(8)

     20 X$="WHERE HAVE ALL THE SOLDIERS      GONE"

     30 A$="soldiers"

     40 ? "soldiers IS AT POSIT";

     50 ? UINSTR(X$,A$)

     EXAMPLE #3

     10 DIM X$(40),A$(8)

     20 X$="ICH BIN EIN BERLINER"

     30 A$="berliner"

     40 ? "berliner IS AT POSIT.";

     50 ? UINSTR(X$,A$,5)

     This command is the Universal INSTR. It searches for the beginning of a sub string.  x$=original string. a$=sub- string.  I=starting position of the search.  With UINSTR you can have lower case, upper case or inverse as the sub string. It does not distinguish between case. You can include or eliminate the starting position of the search. See example #2 and #3.

     FORMAT:  ERR (T)

     EXAMPLE: 20 ? ERR

              90 ? "ERROR # ";ERR;

                 " AT LINE ";ERL

     EXAMPLE #2

     10 CLS:DIM A$(15):?:?

     20 TRAP 90:READ A$

     30 ? A$

     40 GOTO 20

     50 DATA MONDAY,TUESDAY,

        WEDNESDAY

     60 DATA THURSDAY,FRIDAY

     70 DATA SATURDAY,SUNDAY

     90 REM ERROR TRAP ROUTINE

     100 ?:? "ERROR # ";ERR;

         " AT LINE ";ERL

     110 LIST DPEEK(186):REM LIST

         ERROR LINE

     This command lists the last error

number.  It appears to use location 195 which is the same location used by Atari Basic to hold the last Basic error.  If using this location to check for an error, be sure to poke a non-error value

(1) into 195 since Basic only writes error codes to this location.

      FORMAT:  ERL

     EXAMPLE: 20 ? ERL

              90 ? "ERROR # ";ERR;

                 " AT LINE ";ERL

     This command prints the line

number where the last error occurred.

See example #3 at 'ERR"

    **  LOOPS AND SUBROUTINES  **
    -----------------------------

     In TURBO BASIC you call a

subroutine in 3 ways: 1) GOSUB - by

line number. 2) PROCEDURE - by name.

3) USR - by address.

     FORMAT:  REPEAT - UNTIL - EXIT (T)

     EXAMPLE: 10 REPEAT:A=A+1

              20 ? A

              30 UNTIL A=6

     EXAMPLE #2

     10 CLS

     20 A=64

     30 ?:?:?:?

     40 REPEAT:A=A+1

     50 ? CHR$(A);", ";

     60 UNTIL A=90

     This command is a loop. It will repeat itself till the end condition is met. The end condition must be included with the UNTIL statement.  EXIT gets you out of the loop.  Also, see DO-LOOP

     FORMAT:  WHILE - WEND - EXIT (T)

     EXAMPLE: 10 A=10:WHILE A

              20 ? A

              30 A=A-1:WEND

     EXAMPLE #2

     10 CLS:?

     20 A=90:WHILE A:? CHR$(A);

        ", ";

     30 IF A=65 THEN A=1

     40 A=A-1:WEND

     This command is a loop. As long as the WHILE remains NON-ZERO the loop will execute. All statements between WHILE and WEND will be executed. In example #2, A=A-1 is the reverse counter. We terminate the loop before it becomes zero with the IF-THEN. If you go out of this loop with GOTO use the POP to clear the

stack. Example #2 is the same as "FOR X=90 TO 64 STEP-1".  EXIT gets you out of the loop. See DO-LOOP

     FORMAT:  DO - LOOP - EXIT (T)

     EXAMPLE: 10 DO:A=A+1

              20 ? A;", ";

              30 IF A=15O THEN EXIT

              40 LOOP

     The DO-LOOP is an infinte loop unless you use the EXIT command to get out of it. The DO and the LOOP are the 2 parts needed for this loop.  Anything between them will be printed.

     FORMAT:  IF - ELSE - ENDIF

              IF - ENDIF

     EXAMPLE: 90 IF X>3:? "TOO

              LARGE:ELSE:? "OK":

              ENDIF

              10 IF A<7 ? "YES"

              :ENDIF

     EXAMPLE #2

     10 CLS:?:?

     20 INPUT "GIVE ME A NUMBER";A

     30 IF A>5

     40    PRINT A

     50    GO# TOO_BIG

     60 ELSE

     70    PRINT A

     80    GO# A_IS_OK

     90 ENDIF

     100 # TO_BIG:? "THIS NUMBER

         IS TOO LARGE":?:GO TO 20

     110 # A_IS_OK:? "THIS NUMBER IS

         JUST RIGHT"

     120 END

     EXAMPLE #3

     10 CLS:?:?:DIM A$(1)

     20 INPUT "IS THIS PORRIDGE

        OK(Y/N)";A$

     30 IF A$(1,1)="N":GO# TOO_HOT:

        ELSE:GO# JUST_RIGHT:ENDIF

     40 # TOO_HOT:?:? "THIS PORRIDGE

        IS TOO HOT SAID GOLDILOCKS"

        :END

     50 # JUST_RIGHT:? :? "THIS IS

        JUST RIGHT SAID GOLDILOCKS"

     This command has NO "THEN" in it.

It can be used with an ELSE or not.  If you use it on one line be sure to have a colon before and after the ELSE (:ELSE:). This command must be terminated with an :ENDIF. The ENDIF has a colon in front of it if it is used on one line. See example #3 line 30.

     FORMAT:  PROC - ENDPROC -(EXEC) (T)

     EXAMPLE: 10 EXEC START_PROG

              20 END

              30 PROC START_PROG

              40 ? "HELLO"

              50 ENDPROC

     EXAMPLE #2:

     10 CLS:?:?

     20 EXEC TOO_HARD

     30 EXEC TOO_SOFT

     40 EXEC JUST_RIGHT

     50 END

     60 --

     70 PROC TOO_HARD

     80 ? "GOLDILOCKS SAID THIS BED

        IS TOO HARD!":?

     90 ENDPROC

     100 --

     110 PROC TOO_SOFT

     120 ? "GOLDILOCKS SAID THIS

         BED IS TOO SOFT!":?

     130 ENDPROC

     140 --

     150 PROC JUST_RIGHT

     160 ? "GOLDILOCKS SAID THIS IS

         JUST RIGHT!"

     170 ENDPROC

     This command is used like GOSUB

except we use a variable label name.

Just like GOSUB you should have an

END statement somewhere in your

program. The PROC stand for procedure

and is called by the EXEC (execute)

command. The EXEC looks for the PROC

name and then executes it. The

command is terminated with ENDPROC.

Anything between PROC and ENDPROC

will be executed.

          **  GRAPHICS  **
          ----------------

     To get a clear understanding of the various graphics modes, colors, screen positions and text windows etc.  see the book "YOUR ATARI COMPUTER" by Lon Poole et al. published by Osbrone/Mc Graw-Hill

     FORMAT:  CIRCLE x,y,r

     EXAMPLE: 40 CIRCLE 160,96,20

     EXAMPLE #2

     10 GRAPHICS 8+16

     20 COLOR 1

     30 FOR X = 1 TO 90 STEP 3

     40 CIRCLE 160,96,X

     50 NEXT X

     60 GOTO 20

     EXAMPLE #3

     10 GRAPHICS 8+16

     20 COLOR 1

     30 FOR X=20 TO 300 STEP 20

     40 CIRCLE X,96,80

     50 NEXT X

     60 GOTO 20

     This command draws a circle. The graphic mode has to be stated in the program. x and y are the center position of the circle. r is the radius of the circle.

     FORMAT:  CIRCLE x,y,r,r2

     EXAMPLE: 30 CIRCLE 160,96,50,70

     EXAMPLE #2:

     10 GRAPHICS 8+16

     20 COLOR 1

     30 FOR X=1 TO 75 STEP 3

     30 CIRCLE 160,96,X,X+25

     40 NEXT X

     50 GOTO 20

     This command is the same as above except we have a vertical radius r2 so we can draw true circles or ellipses. x,y is the center of the circle, r is the radius and r2 is the vertical radius.

     FORMAT:  FCOLOR n (T)

     EXAMPLE: 30 FCOLOR 3

     EXAMPLE #2

     10 GRAPHICS 7+16

     20 COLOR 1

     30 CIRCLE 80,48,45

     40 FCOLOR 3

     50 FILLTO 80,48

     60 FCOLOR 1

     70 FILLTO 80,93

     80 GOTO 20

     FCOLOR is usually used in conjunction with the FILLTO command. FCOLOR selects the fill in color. The number of fill in colors depend on the graphics mode.

     FORMAT:  FILLTO x,y (T)

     EXAMPLE: 70 FILLTO 80,93

     This command fills in a shape with the color selected by the FCOLOR command. It is the same as XIO 18,#6,0,0,"S:". The x and y are the starting position INSIDE the shape.  See example #2 under FCOLOR.

     FORMAT:  PAINT x,y (T)

     EXAMPLE: 40 PAINT 80,48

     EXAMPLE #2:

     10 GRAPHICS 7+16

     20 COLOR 1

     30 CIRCLE 80,48,45

     40 PAINT 80,48

     50 GOTO 20

     This is also a type of fill in command. This is a recursive routine that will fill any closed shape. The x and y have to be the position inside the shape to be filled.

     FORMAT:  TEXT x,y,a$ (T)

     EXAMPLE: 40 TEXT 3,8,A$

     EXAMPLE #2

     10 CLS

     20 DIM A$(4)

     30 A$="WIND"

     40 TEXT 3,8,A$

     This command bit-blocks the text in A$ at position x,y.

  **  CONSTANTS AND ERROR CODES  **
  ---------------------------------

     FORMAT:  %0 %1 %2 %3 (T)

     EXAMPLE: 20  X=%1

              50 FOR T = %1 TO %3:               ? T:NEXT T

     These 4 constants stand for the numbers 0 to 3. They are used like regular numbers. "X=1" uses 10 bytes while "X=%1" uses 4 bytes. When you use these numbers you save on each number 6 bytes. It is good programming to make variables of numbers that are used more than 3 times.

           NEW ERROR CODES
           ---------------

     TURBO-BASIC prints out in English all errors. You do not have to look up the error code numbers anymore. TURBO- BASIC has 8 new error codes not in ATARI- BASIC. The new error codes are form 22 to 30

     FORMAT:  Error - 22 ?NEST (T)

The loops are not properly nested.

     FORMAT:  Error - 23 ?WHILE (T)

A WHILE-WEND loop with no corresponding WHILE.

     FORMAT:  Error - 24 ?REPEAT (T)

An UNTIL with no corresponding REPEAT.

     FORMAT:  Error - 25 ?DO (T)

     A DO-LOOP with no corresponding DO.

     FORMAT:  Error - 26 ?EXIT (T)

     EXIT is outside a loop.

     FORMAT:  Error - 27 ?XPROC (T)

Error in executing a procedure ( There is no END line in your program etc.)

     FORMAT:  Error - 28 ?EXEC (T)

     ENDPROC with no corresponding

     EXEC.

     FORMAT:  Error - 29 ?PROC (T)

     PROC does not exist.

     FORMAT:  Error - 30 ?# (T)

     Variable name label dos not

     exist.

     FORMAT:  Error - 15

     Error 15 has been expanded to

include an UNTIL which is in a REPEAT- UNTIL loop and which has been deleted by mistake.

    **  ARITHMETIC AND LOGIC  ** 
    ----------------------------

     FORMAT:  HEX$(n)

     EXAMPLE: 20 ? HEX$(32)

     EXAMPLE #2

     10 CLS:?:?

     20 INPUT "GIVE ME A DECIMAL

        NUMBER";A

     30 ? "DECIMAL ";A;" = HEX ";

        HEX$(A)

     40 ?:GOTO 20

This command takes a decimal number and converts it into a HEXADECIMAL number

     FORMAT:  DEC(a$)

     EXAMPLE: 20 A$="FF":? DEC(A$)      EXAMPLE #2

     10 CLS:?:?: DIM A$(8)

     20 INPUT "GIVE ME THE HEX

        NUMBER";A$

     30 ? "HEX ";A$;" = DECIMAL ";

        DEC(A$)

     40 ?:GOTO 20

This command changes a HEXADECIMAL number to a decimal number. The HEX number must first be put into a string.

     FORMAT:  n DIV i

     EXAMPLE: 20 A=9:B=4:? A DIV B

              EXAMPLE #2

     10 CLS:?:?

     20 INPUT "GIVE ME A NUMBER";A

     30 INPUT "GIVE ME A SMALLER

        NUMBER";B

     40 Q=A DIV B:R=A MOD B

     50 ? A;"/";B;" = ";

     60 ? Q;" REMAINDER ";R

This command gives you the whole number part of a division or fraction. Example 9/4=2 R 1. This command prints the 2.

     FORMAT:  n MOD i

     EXAMPLE: 20 A=9:B=4:? A MOD B

This command returns the remainder of a division as  a whole number. Example 9/4=2 R 1 This command will print 1. See example #2 above.

     FORMAT:  FRAC(a)

     EXAMPLE: A=22/7:? FRAC(A)

     EXAMPLE #2

     10 CLS:?:?

     20 A=22/7

     30 ? "THE FRACTION 22/7 = ";A

     40 ? "THE DECIMAL PART OF

        22/7 = ";FRAC(A)

This command prints the decimal part of a fraction ONLY. Example 22/7=3.142857... This command prints 142857...

     FORMAT:  TRUNC(a)

     EXAMPLE: A=22/7:? TRUNC(A)

     EXAMPLE #2

     10 CLS:?:?

     20 A=22/7

     30 ? "THE FRACTION 22/7 =";A

     40 ? "THE INTEGER PART =";

        TRUNC(A)

This command prints the Integer part of a fraction ONLY. Example 22/7=3.142857 it will print only 3.

     FORMAT:  $nnnn

     EXAMPLE: 20 POKE $8000

     EXAMPLE #2

     10 CLS:?:?

     20 FOR T = $0F TO $FF

     30 ? T;", ";

     40 NEXT T

     This command allows you to use hex. numbers but they are converted to decimal numbers. In example #2 line 20 is converted to 'FOR T = 15 TO 255.

        **  BOOLEAN LOGIC  **
        ---------------------

     The &, !, EXOR, are used in BOOLEAN logic. They are usually used to compare 2 bits. The bits have to be positive. Negative numbers will give you an error. Each program will print out a table for you of the result of comparing 2 bits

     FORMAT:  n & i

     EXAMPLE:

     10 CLS:?:?

     20 ?,"BOOLEAN 'AND' (&)"

     30 ?

     40 ? "BIT ONE","BIT TWO", "RESULT"

     50 ? "1     &","1 =",1&1

     60 ? "0     &","1 =",0&1

     70 ? "1     &";"0 =",1&0

     80 ? "0     &";"0 =",0&0

     This command is the BOOLEAN 'AND'

     FORMAT:  n ! i

     EXAMPLE:

     10 CLS:?:?

     20 ? ,"BOOLEAN 'OR' (!)"

     30 ?

     40 ? "BIT ONE","BIT TWO",

        "RESULT"

     50 ? "1     !","1 =",1!1

     60 ? "0     !","1 =",0!1

     70 ? "1     !","0 =",1!0

     80 ? "0     !","0 =",0!0

     This command is the BOOLEAN 'OR'

     FORMAT:  n EXOR I

     EXAMPLE:

     10 CLS:?:?

     20 ? " BOOLEAN 'EXCULSIVE OR' (EXOR)

     30 ?

     40 ? "BIT ONE","BIT TWO","RESULT"

     50 ? "1  EXOR";"1 =",1 EXOR 1

     60 ? "0  EXOR";"1 =",0 EXOR 1

     70 ? "1  EXOR","0 =",1 EXOR 0

     80 ? "0  EXOR","0 =",0 EXOR 0

     This command is the BOOLEAN "Exclusive Or'

    **  SELF BOOTING PROGRAMS  **
    -----------------------------

If you want to have a program that runs itself upon booting up the disk name it 'AUTORUN.BAS'. TURBO BASIC looks for a BASIC file by this name when the boot process is in operation

            **  NOTES  **
            -------------

Variables, Procedure Names and labels may contain the underscore (_). Now you can combine several words into a name that tells you WHAT and WHERE    it goes. This is a powerful feature to make your

programs more legible.

     EXAMPLES:

     10 TRAP 200

     10 TRAP #END_OF_FILE

     60 X = 65

     60 # LAST_CHAR = 65

     90 IF A$(1,1) = "Y" THEN 60

     90 IF A$(1,1) = "Y" THEN #         FETCH_LABEL

            DOUBLE QUOTES
            -------------

Double quotes are allowed in print statements. Whenever you want a quote in a print statement put in double quotes.

EXAMPLE:

30 ? "SHE SAID ""GOODBYE"" AND SLAMMED THE DOOR"

    MULTIPLE LINE IF-ELSE-ENDIF
    ----------------------------

You can use several lines between IF and ELSE and ENDIF. See example #2 of IF-ELSE-ENDIF on page 24

               ARRAYS
               ------

You can have only NUMBER arrays.  Just like ATARI BASIC, TURBO-BASIC does not allow string arrays. Number arrays can be one or two dimensional. You must DIM each array.  One dimensional array 'DIM A(5)'. Two dimensional arrays 'DIM A(5,8)'.

          ***************** 
          *               * 
          * PUBLIC DOMAIN * 
          *               * 
          *****************
     I am collecting TURBO-BASIC utilities to be put into the Public Domain. Your help is appreciated. If you have a utility to share please send it to the author (I prefer it on a disk if possible).

     This documentation is in the Public Domain and may be copied by anyone!

     The author of this documentation

apologizes for any omission or mistakes - they were accidental.

     If you cannot get this disk through any Public Domain source please send $3.00 to the author to cover the expense of the disk, the mailer and postage.

     If you have any comments or

suggestion about this documentation

- they are welcome. Please write to:

            RON FETZER

            22 MONACO AVE

            ELMONT, N.Y. 11003 USA

